Dataplot Vol 2 Vol 1

# VARIATIONAL DISTANCE

Name:
VARIATIONAL DISTANCE (LET)
Type:
Let Subcommand
Purpose:
Given a vector of counts, compute the difference from uniformity based on the variational distance.
Description:
In spatial analysis, it is sometimes desired to determine if the points in the given space are consistent with a uniform distribution. One such measure is based on the variational distance which is defined as

$$d = \frac{1}{2} \sum_{k=0}^{\infty}{|P(Unif = k) - P(data=k)|}$$

Given that the points have been converted to a set of N counts, Xk, this formula becomes

$$d = \frac{1}{2} \frac{\sum_{k=1}^{N}{|\frac{1}{N} - X_{k}|}} {\sum_{k=1}^{N}{X_{k}}}$$

The value of the variational distance is between zero and one with values closer to zero indicating greater consistency with a uniform distribution.

Syntax:
LET <a> = VARIATIONAL DISTANCE <y>
<SUBSET/EXCEPT/FOR qualification>
where <y> is the response variable;
<a> is a parameter where the computed statistic is saved;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.
Examples:
LET A = VARIATIONAL DISTANCE Y1
LET A = VARIATIONAL DISTANCE Y1 SUBSET TAG > 2
Note:
The relative dispersion index is a scaled version of the variational distance. Enter HELP RELATIVE DISPERSION INDEX for details.
Note:
Dataplot statistics can be used in a number of commands. For details, enter

Default:
None
Synonyms:
None
Related Commands:
 RELATIVE DISPERSION INDEX = Compute the relative dispersion index of a variable containing a set of counts. UNIFORM CHISQUARE STAT = Compute the chi-square statistic for uniformity for a variable containing a set of counts. GOODNESS OF FIT = Perform a goodness of fit test.
Reference:
Kashiwagi, Fagan, Douglas, Yamamoto, Heckert, Leigh, Obrzut, Du, Lin-Gibson, Mu, Winey, Haggennueller (2007), "Relationship between dispersion metric and properties of PMMA/SWNT nanocomposites", Polymer Journal, Vol. 48, pp. 4855 - 4866.
Applications:
Spatial Statistics
Implementation Date:
2014/3
Program:

LET Y  = UNIFORM RANDOM NUMBERS FOR I = 1 1 1000
LET Y1 X1 = BINNED Y
LET Y  = NORMAL RANDOM NUMBERS FOR I = 1 1 1000
LET Y2 X2 = BINNED Y
LET Y  = EXPONENTIAL RANDOM NUMBERS FOR I = 1 1 1000
LET Y3 X3 = BINNED Y
LET A1 = VARIATIONAL DISTANCE Y1
LET A2 = VARIATIONAL DISTANCE Y2
LET A3 = VARIATIONAL DISTANCE Y3
SET WRITE DECIMALS 4
PRINT A1 A2 A3

The following output is generated.
 PARAMETERS AND CONSTANTS--

A1      --         0.0747
A2      --         0.3924
A3      --         0.5181


NIST is an agency of the U.S. Commerce Department.

Date created: 06/23/2014
Last updated: 06/23/2014